
PLMP - Point-Line Minimal Problems in Complete Multi-View Visibility
Suplementary Material

Timothy Duff
School of Mathematics, Georgia Tech

Kathlén Kohn
KTH

Anton Leykin
School of Mathematics, Georgia Tech

Tomas Pajdla
CIIRC - Czech Technical University in Prague*

Here1 we present additional details for paper [4]. In particu-
lar, we provide more details about our notation, explain ba-
sic concepts from algebraic geometry used, and implemen-
tation details including commented implementations of typ-
ical examples of dominance checking and symbolic as well
as numeric degree computation. We also relate our 32003
problems to the classical formulation based on calibrated
homography [5]. The code of all examples is available at

https://github.com/timduff35/PLMP

1. Notation and concepts
We use nomenclature from [7] for basic concepts in geom-
etry of computer vision. See [2] for the fundamentals of
algebraic geometry, including Gröbner bases.

SO(3) stands for the special orthogonal group, i.e. ro-
tations, defined algebraically as 3 × 3 matrices R such that
RR⊺ = I and det(R) = 1. We note that the dimension of
SO(3) is three.

For u in R3, the skew-symmetric matrix [u]× in R3×3

represents the cross product with u in R3, i.e. [u]× v = u×v
for all v in R3.

Points in space are in the projective space P3, whereas
image points are in P2. So points are represented by homo-
geneous coordinates.

We sometimes refer to the concept of algebraic varieties.
A projective variety is the common zero set of a system of
homogeneous polynomial equations. If the polynomials are
defined in N + 1 homogeneous variables, the projective va-
riety lives in PN . Similarly, an affine variety is the common
zero set of a system of polynomial equations which are not
necessarily homogeneous. If the polynomials are defined in

N variables over the ground field F, the affine variety lives
in FN . For any subset S of either PN or FN , the Zariski clo-
sure S of S is the smallest projective, resp. affine, variety
containing S.

The Grassmannians G1,3 and G1,2, which are the sets of
all lines in P3 and P2, respectively, are examples of smooth
projective varieties. The Grassmannian G1,2 of lines in P2

is isomorphic to P2 as every line in the projective plane is
uniquely defined by a single linear equation with three ho-
mogeneous coordinates as coefficients. The Grassmannian
G1,3 of lines in P3 can be seen as a projective variety via its
embedding into P5 defined by its Plücker coordinates.

A line in P3 is defined by two linear equations. We write
the homogeneous coefficients of these two equations as the
two rows of a 2 × 4 matrix [ a0 a1 a2 a3

b0 b1 b2 b3 ]. The six max-
imal minors of this matrix form the Plücker coordinates
of the line in P3: pij = aibj − ajbi. Every line in P3 is
uniquely defined by its six Plücker coordinates. Moreover,
the Plücker coordinates of a line in P3 satisfy the equation
p01p23 − p02p13 + p03p12 = 0. In addition, every projective
tuple (p01 ∶ p02 ∶ p03 ∶ p12 ∶ p13 ∶ p23) ∈ P5 satisfying
the equation p01p23 − p02p13 + p03p12 = 0 is the Plücker
coordinates of a line in P3.

This shows that the Grassmannian G1,3 is the zero set
of the polynomial equation p01p23 − p02p13 + p03p12 = 0
in P5. Furthermore, the Plücker coordinates of a line in P3

serve as homogeneous coordinates defining the line. So we
can represent both points and lines in P2 as well as P3 by
homogeneous coordinates.

A variety X is said to be irreducible if it is not the union
of two proper subvarieties, i.e. if it cannot be written asX =
X1 ∪X2 where X1 and X2 are non-empty subvarieties of
X which are not equal to X . For instance, Grassmannians
are irreducible.

Throughout our article, we consider an arbitrary ground
field F unless explicitly specified. For different purposes we
use different fields. For instance, the minimal problems we
study are clearly defined over the field R of real numbers.

When setting up systems of polynomial equations, coeffi-
cients originate from the field Q of rational numbers. So-
lutions of the equations are in the field C of complex num-
bers. We carry out symbolic computations in a finite field
Zp for a prime p for the sake of exactness and computa-
tional efficiency. Numerical algorithms use floating point to
approximate complex numbers.

2. Additional details for 32003 problem

Here we further investigate 32003 problem and re-
late it to the classical formulations. The code is in
"PLMP-3200_3.m2" file.

Problem 32003 has five points on two lines with one
point in common to both lines. Hence, all five points are in a
single 3D plane. This corresponds to the relative calibrated
camera pose from homography [5, 9]. It first computes a ho-
mography between the two images using four point corre-
spondences. The fifths correspondence is determined by the
four and automatically satisfies the homography. Having
four correspondences, a single solution, A ∈ R3×3 viewed
as in P9, is obtained and then decomposed into multiple
possible rotations, translations, normals and a scales.

The decomposition problem can be formulated as fol-
lows [5]

A = dR + T NT

with rotation R, translation T , normal N and scale d.
The decomposition of a homography matrix A, which

was developed in [5], is based on the SVD. It generates 8,
resp. 4, solution in a generic situations with ∣∣N ∣∣2 = 1, resp.
N3 = 1, dehomogenization. The following Macaulay2 code
shows that the degree over the complex numbers is 24, resp.
12, in a generic situations with ∣∣N ∣∣2 = 1, resp. N3 = 1.

Rng = QQ[r (1,1)..r (3,3),t 1..t 3,n 1..n 3,d]
R = transpose(genericMatrix(Rng,r (1,1),3,3))
T = genericMatrix(Rng,t 1,3,1)
N = genericMatrix(Rng,n 1,3,1)
A = random(QQ^3,QQ^3)
I = ideal(A - d*R-T*transpose(N)) +

ideal(transpose(R)*R-id (QQ^3)) +
ideal(det(R)-1) +
ideal(transpose(N)*N-1)

dim I, degree I
o7 = (0, 24)

I = ideal(A - d*R-T*transpose(N)) +
ideal(transpose(R)*R-id (QQ^3)) +
ideal(det(R)-1) +
ideal(N 3-1)

dim I, degree I
o7 = (0, 12)

We used a linear constraint to dehomogenize when com-
puting the degree of 32003 problem over the complex num-
bers and obtained its degree equal to 12, which is in agree-
ment with the above homography formulation.

3. Implementation details
In this section, we give describe the Macaulay2 [6] imple-
mentation of our computations through a series of guided
examples. Each subsection provides commentary for a ded-
icated Macaulay2 file:

3.1 "example-2111_1-unrolled.m2"

3.2 "example-2111_1.m2"

3.3 "example-2111_1-jacobian.m2"

3.4 "example-2111_1-numerical.m2"

Besides these examples and setup code providing the core
functionality, there are several scripts which automate the
computations reported in the main text.

3.1. Computing the degree symbolically “from
scratch"

We begin by setting a ground field FF to be the finite field
Z10007:

FF = ZZ/10007

We define point data in P2 in three views over Z10007 con-
sistent with the problem “21111”). This is represented as a
list of three matrices in Macaulay2, which can be viewed in
an output buffer using the netList command:

i2 : P = {matrix{{-2639,-4936,1789},
{2653,-591,-643},
{1,1,1}},

matrix{{-3868,-1776,3174},
{3669,-4143,-1982},
{1,1,1}},

matrix{{-1889,-1604,4629},
{-473,-4513,-4210},
{1,1,1}}};

i3 : netList P

+---------------------+
o3 = || -2639 -4936 1789 | |

|| 2653 -591 -643 | |
|| 1 1 1 | |
+---------------------+
|| -3868 -1776 3174 ||
|| 3669 -4143 -1982 ||
|| 1 1 1 ||
+---------------------+
|| -1889 -1604 4629 ||
|| -473 -4513 -4210 ||
|| 1 1 1 ||
+---------------------+

The first two columns of each matrix in P were randomly
generated, while the last column is (projectively) a random
Z10007-linear combination of the first two. This reflects the
dependence of the third point on the first two in each view.

In general, we encode point-line problems by specifying
the number of visible lines, the number of ghost lines, and,
for each point, a list of lines passing through it. For problem
“21111” this is accomplished as follows:



D = (3,2,{{1,2},{1,3},{1,4}})
nLines = D#0 + D#1
lineIncidences = D#2

The list L contains random line matrices (homogeneous
coordinates of lines as in rows) consistent with P and the
encoding D:

i8 : netList L

+---------------------+
o8 = || 107 4376 3187 ||

|| 3897 -4638 2998 ||
|| -1009 2785 -4328 ||
|| 2883 -1540 1031 ||
|| -1517 -713 3879 ||
+---------------------+
|| 3783 -1437 275 ||
|| 2926 -687 -1310 ||
|| -2582 -2466 1236 ||
|| -2721 -1085 -1127 ||
|| 3072 4259 1727 ||
+---------------------+
|| -1563 -4868 -1852 ||
|| -968 -960 -1036 ||
|| -3676 382 1454 ||
|| 1795 2177 -4909 ||
|| 2696 -1409 1206 ||
+---------------------+

We may verify that the first row of each matrix in L rep-
resents a free line, that the second row gives a line through
all points, the third rows gives a “pinned line” through the
first point, and the fourth and fifth rows give ghost lines
through the second and third points, respectively.

i9 : netList apply(L,P,(l,c)-> l*c)

+---------------------+
o9 = || 2418 996 2676 | |

|| 0 0 0 | |
|| 0 -2170 2336 | |
|| -4650 0 4640 | |
|| 4176 -2363 0 | |
+---------------------+
|| -799 -4310 -4744 ||
|| 0 0 0 ||
|| 0 3153 -4110 ||
|| -1742 0 -2575 ||
|| 2884 -3026 0 ||
+---------------------+
|| -456 -2638 -1974 ||
|| 0 0 0 ||
|| 0 873 57 ||
|| -2291 0 -422 ||
|| -1987 4218 0 ||
+---------------------+

Now, we define a polynomial ring over FF with indeter-
minates for the 11 unknown camera parameters:

R = FF[r_(1,1), r_(1,2), r_(1,3),
r_(2,1), r_(2,2), r_(2,3),
t_(1,1), t_(2,1),t_(1,2), t_(2,2), t_(2,3)]

The next line defines a list Rs to be a list representing each
of the camera matrices’ rotations.

Rs = {matrix{{1,0,0},{0,1,0},{0,0,1}},
matrix{{ -r_(1,1)^2-r_(1,2)^2+r_(1,3)^2+1,

-2*r_(1,2)*r_(1,3)+2*r_(1,1),
2*r_(1,1)*r_(1,3)+2*r_(1,2) },

{ -2*r_(1,2)*r_(1,3)-2*r_(1,1),
-r_(1,1)^2+r_(1,2)^2-r_(1,3)^2+1,
-2*r_(1,1)*r_(1,2)+2*r_(1,3) },

{ 2*r_(1,1)*r_(1,3)-2*r_(1,2),
-2*r_(1,1)*r_(1,2)-2*r_(1,3),
r_(1,1)^2-r_(1,2)^2-r_(1,3)^2+1 }},

matrix{{ -r_(2,1)^2-r_(2,2)^2+r_(2,3)^2+1,
-2*r_(2,2)*r_(2,3)+2*r_(2,1),
2*r_(2,1)*r_(2,3)+2*r_(2,2)},

{ -2*r_(2,2)*r_(2,3)-2*r_(2,1),
-r_(2,1)^2+r_(2,2)^2-r_(2,3)^2+1,
-2*r_(2,1)*r_(2,2)+2*r_(2,3) },

{ 2*r_(2,1)*r_(2,3)-2*r_(2,2),
-2*r_(2,1)*r_(2,2)-2*r_(2,3),
r_(2,1)^2-r_(2,2)^2-r_(2,3)^2+1

}}};

Similarly, we may define a list of camera translations,

ts={matrix{{0},{0},{0}},
matrix{{t_(1,1)},{t_(1,2)},{1}},
matrix{{t_(2,1)},{t_(2,2)},{t_(2,3)}}};

and a list of camera matrices

C = apply(Rs,ts,(R,t)->R|t);

The polynomials defining each line correspondence (LC)
and common point constraint (CP) will depend on C and
certain submatrices from L. The following functions return
ideals generated by these polynomials:

cl = (C,L’) -> minors(3,
matrix apply(C,L’,(c,l)->{l*c}),
Strategy=>Cofactor);

cp = (C,L’) -> minors(4,
matrix apply(C,L’,(c,l)->{l*c}),
Strategy=>Cofactor);

We let I be the ideal of camera poses consistent with L,
which may be constructed as follows:

Icl = sum(nVisibleLines,
i->cl(C,L/(l->l^{i})));

Icp = sum(lineIncidences,
LINES->cp(C,L/(l->l^LINES)));

I = Icl + Icp;

At this point we must note that the matrices defined in
Rs are “scaled” so that the denominators appearing in the
usual Cayley parameterization of SO(3)—namely, d1 =
r22,1 + r22,2 + r22,3 + 1 and d2 = r23,1 + r23,2 + r23,3 + 1—do
not appear. This scaling has the effect of introducing spu-
rious solutions to the determinantal equations where these
denominators may be zero. A standard technique for erad-
icating these spurious solutions is via introducing an auxil-
iary variable z and equation z (d1 d2) + 1 enforcing d1 ≠ 0
and d2 ≠ 0:

Rz = FF[z, gens R,
MonomialOrder=>{Eliminate 1}]

Iz = sub(I,Rz) + ideal(1+z*
(r_(2,1)^2+r_(2,2)^2+r_(2,3)^2+1)*
(r_(1,1)^2+r_(1,2)^2+r_(1,3)^2+1));



We may compute a Gröbner basis for the ideal Iz using
Macaulay2’s implementation of the F4 algorithm.

gbIz = groebnerBasis(Iz, Strategy => "F4");

The monomial order for the ring Rz is a block-wise graded
reverse lexicographical ordering that eliminates the auxil-
iary variable z. It follows [2] that the original ideal I is
generated by the Gröbner basis elements which do not con-
tain this variable.

gbI = selectInSubring(1,gbIz);

We get the dimension and degree of I by passing to the
initial ideal.

i23 : inI = ideal sub(leadTerm gbI,R);

o23 : Ideal of R

i24 : dim inI

o24 = 0

i25 : degree inI

o25 = 40

3.2. Computing the degree symbolically using setup
code

The following Macaulay2 code replicates the entire
Gröbner basis computation outlined in the previous section.
Note that a seed for the random number generator has been
set to ensure a reproducible result.

setRandomSeed 0;
m = 3;
FF = ZZ/nextPrime 10000;
isParametric = false;
D = (3,2,{{1,2},{1,3},{1,4}});
needs "problem-builder.m2";
gbIz = groebnerBasis(Iz, Strategy => "F4");

The script "problem-builder.m2" and its depen-
dencies automate tasks such as building equations and gen-
erating point-line data. It assumes several global variables
have been defined:

• m — the number of cameras

• FF — the ground field

• isParametric — a Boolean determining how
point-line data are represented. A value of false
indicates that the data are set up randomly over the
ground field. A value of true means that point-line
data are defined in terms of parametric indeterminates,
to be specialized later at some fabricated point as in the
minimality check described in the next section.

• D — an encoding of the problem as described in the
previous section.

3.3. Minimality check via Jacobian

We now show how to check minimality without comput-
ing degrees for “21111.”

setRandomSeed 0;
m = 3;
FF = ZZ/nextPrime 10000
isParametric = true;
D = (3,2,{{1,2},{1,3},{1,4}});
needs "problem-builder-matrices.m2"
matrices = pointMatrices | lineMatrices;

As before, the CP and LC matrices are defined over
a polynomial ring in 11 indeterminates. However, the co-
efficient ring is itself a polynomial ring in 27 indetermines,
representing the data defining an instance of the problem.

i8 : (numgens R, numgens coefficientRing R)

o8 = (11, 27)

As in the pseudocode for the minimality check, we fabri-
cate a parameter point satisfying the LC and CP equations:

i9 : xy = fabricatedVectorFLPQ D;

1 38
o9 : Matrix FF <--- FF

i10 : netList(lineMatrices/(
m->ker transpose sub(m,xy)))

+---------------+
o10 = |image | -2428 ||

| | -1877 ||
| | 1 ||
+---------------+
|image | -1975 ||
| | 4507 ||
| | 1 ||
+---------------+
|image | 2657 ||
| | -3077 ||
| | 1 ||
+---------------+

i11 : netList(pointMatrices/(
m->ker sub(m,xy)))

+---------------+
o11 = |image | -2511 ||

| | 417 ||
| | -4137 ||
| | 1 ||
+---------------+
|image | -3306 ||
| | 2306 ||
| | 1800 ||
| | 1 ||
+---------------+
|image | 3529 ||
| | 2429 ||
| | -1474 ||
| | 1 ||
+---------------+



The function goodMinors then checks minimality by
computing a subset of the equations whose Jacobian at xy
with respective to the 11 camera parameters has maximal
rank.

i12 : gm = goodMinors(pointMatrices,
lineMatrices,xy);

i13 : #gm

o13 : 11

3.4. Numerical computations of degree

As an additional check, the degrees of minimal prob-
lems with 2 and 3 cameras were re-computed using mon-
odromy. [3] This is a general, randomized technique for
solving parametric polynomial systems F (c, y) = 0 for
generic parameter values y. We also used monodromy to
compute the degrees of problems with 4 and 5 cameras.
For the minimal problem with 6 cameras, monodromy did
not terminate after several days, but computed more than
450,000 solutions. Rather than provide detailed pseu-
docode, we outline the salient features of this approach:

• Starting from a generic point (c0, y0) ∈ Inc′ (cf. main
text) with F (c0, y0) = 0, known solutions are numer-
ically continued along a fixed set of paths h0, . . . hj ,
where each hi ∶ [0,1] → Yp,l,I,m is a random path
connecting the solutions of F (⋅, y0) = 0 with those of
some other fixed instance F (⋅, y1) = 0

– Here, “random path” is meant in the sense that
each hi(0) (resp. hi(1)) is randomly generated
point defining a problem instance with the same
solutions as y0 (resp. y1.) For example, we might
take hi(0) = γ y0 for some random constant γ ∈
C, yielding the well-known γ-trick of [10].

– Continuation along some path from t = 0 to 1
followed by continuation from 1 to 0 induces a
permutation of the solutions of F (⋅, y0) = 0. The
group of all such permutations acts transitively
by the irreducibility of Inc′—thus all solutions
may be discovered starting from one.

• The algorithm maintains a set of correspondences be-
tween solutions along each path hi, and attempts the
above continuation step until no other correspondences
may be established.

– This stabilization-based stopping criterion comes
with no theoretical guarantee of correctness.
Heuristic arguments suggest that all solutions are
found with very high probability, even for small
j and problems of moderate-to-large degree. [3]
[1] Multiple correct runs for the problems in 2
and 3 views support this thesis, increasing our

confidence in the reported results. Nevertheless,
we explicitly mark in our table of results the
degrees which were only computed numerically
with a “*”, a practice originating from [8].

Computations were performed using the Macaulay2
package MonodromySolver. At the time of writing, this
package is available in the current release of Macaulay2
(version 1.13.) However, the computations made use of fea-
tures not available in the currently released package:

• To guard against the possibility of path-jumping
(which could result in extraneous solutions), the ranks
of all CP and CL matrices were computed numeri-
cally for each newly discovered solution by threshold-
ing singular values less than 10−4. If any CP or CL
matrix did not yield the expected ranks of 3 or 2, the
solution was thrown out and not used to generate fur-
ther correspondences.

• To speed up time-intensive polynomial evalua-
tion, MonodromySolver was linked to the pack-
age SLPexpressions, which allows for custom
straight-line program representations of polynomial
systems.

These enhancements will be incorporated into a future
version of the package.

The output of the main function monodromySolve in-
cludes solution data for a chosen parameter point yi, which
may be written to file. This solution data may be used as a
start system for solving an instance of the same problem via
homotopy continuation. We illustrate this for the minimal
problem “21111.”

We begin by loading the setup script
"numerical-problem-builder.m2".

setRandomSeed 0
m = 3
D = (3,2,{{1,2},{1,3},{1,4}})
Jpivots = {0, 1, 4, 5, 8, 9, 14, 29,

33, 44, 48, 57, 58, 59}
RERUNMONODROMY = false;
needs "numerical-problem-builder.m2"

The global variable Jpivots is a list indexing a square
subsystem of the LC and CP equations. These equations
are the same except that rotations and translations are rep-
resented in homogeneous coordinates (i.e. quaternion pa-
rameterization of SO(3))—yielding an extra three variables
and equations. The square subsystem, depending on para-
metric indeterminates y ∈ C62 as well as c ∈ C14, is an
object of type GateMatrix which the setup script assigns
to the variable F. We may load the results of a previous
monodromy run for this problem and verify that the starting
solutions are valid:



i7 : (yStart, cStarts) =
readStartSys "2111_1-start";

i8 : max(cStarts/(c->
norm evaluate(F,c||yStart)))

o8 = 2.49953543586694e-12
o8 : RR (of precision 53)

We now consider the problem of reconstructing relative
camera poses

i13 : netList rotations23

+-------------------------------+
o13 = || .100226 -.858789 .502431 ||

|| .991267 .129682 .0239204 ||
|| -.0856987 .495646 .864287 ||
+-------------------------------+
|| -.246694 .614271 .749542 | |
|| .828175 -.268025 .492228 | |
|| .503258 .742182 -.442603 | |
+-------------------------------+

i14 : netList translations23

+-------------+
o14 = || .455949 ||

|| .00471457 ||
|| .771746 ||
+-------------+
|| .74002 | |
|| .20217 | |
|| .774737 | |
+-------------+

from line data in each view

i15 : netList L

+------------------------------+
o15 = || -.729481 -.373996 .5727 | |

|| -.617925 -.259526 .742169 | |
|| -.33284 -.499815 .799626 | |
|| -.441933 -.298528 .845917 | |
|| -.886353 -.145719 .439481 | |
+------------------------------+
|| -.568543 .804717 -.170851 ||
|| -.685669 .639348 -.347982 ||
|| .881415 -.405026 .243023 ||
|| .490952 -.819545 .295486 ||
|| .158324 -.969189 .188695 ||
+------------------------------+
|| -.518733 .16617 .838632 | |
|| -.61258 .0842219 .785909 | |
|| -.468687 -.259022 .844535 | |
|| -.595602 -.232359 .768939 | |
|| -.589568 .295501 .751724 | |
+------------------------------+.

The derived (parameter, solution) pair is defined by
respective variables yTarget, cTarget. The
starting solutions are numerically continued to so-
lutions of F specialized at using core functions for
NumericalAlgebraicGeometry in Macaulay2.

PH’ = specialize(PH,yStart||yTarget);
cTargets = trackHomotopy(PH’, cStarts);

The following code filters the target solutions cTargets
for the ground-truth solution y.
cTargets2 = cTargets/(c-> (

R2params = take(c.Coordinates,{0,3});
Q2R(R2params,Normalized=>true)));

i=minPosition(
cTargets2/(R2->
norm(R2-rotations23#0)));

y=(cTargets#i).Coordinates;

We recover the original camera poses:
i23 : Q2R(take(y,{0,3}),Normalized=>true)

o23 = | .100226 -.858789 .502431 |
| .991267 .129682 .0239204 |
| -.0856987 .495646 .864287 |

3 3
o23 : Matrix CC <--- CC

53 53

i24 : Q2R(take(y,{4,7}),Normalized=>true)

o24 = | -.246694 .614271 .749542 |
| .828175 -.268025 .492228 |
| .503258 .742182 -.442603 |

3 3
o24 : Matrix CC <--- CC

53 53

i25 : transpose matrix take(y,{8,13})

o25 = | .455949 |
| .00471457 |
| .771746 |
| .74002 |
| .20217 |
| .774737 |

